

Estimation of nutrient requirements using broken-line regression analysis¹

K. R. Robbins,^{*2} A. M. Saxton,^{*} and L. L. Southern[†]

^{*}Department of Animal Science, University of Tennessee, Knoxville 37996-4588;
and [†]Louisiana State University Agricultural Center, Baton Rouge 70803

ABSTRACT: We evaluated and compared various broken-line regression models and SAS (SAS Inst. Inc., Cary, NC) procedures for estimating nutrient requirements from nutrient dose response data. We used the SAS (Version 9) procedures NLIN and NLMixed and the response data of Parr et al. (2003), who evaluated the isoleucine requirement of growing swine. The SAS NLIN was used to fit 2 different broken-line regression models: a simple 2 straight-line, one-breakpoint model and a quadratic broken-line model in which the response below the single breakpoint was quadratic; there was a plateau above the breakpoint. The latter was fit using 2 different approaches in NLIN. We also used

SAS NLMixed to fit 3 different broken-line models: the 2 straight-line, one-breakpoint model that included a random component for the plateau; the quadratic broken-line model that included a random component for the plateau; and the quadratic broken-line model that included random components for both the plateau and the slope of the curve below the requirement. The best fit (greater adjusted R²; least log likelihood) was achieved using SAS NLMixed and the quadratic model with a random component for asymptote included in the model. Model descriptions, SAS code, and output are presented and discussed. Additionally, we provide other examples of possible models and discuss approaches to handling difficult-to-fit data.

Key words: nonlinear regression, nutrient requirement

©2006 American Society of Animal Science. All rights reserved. J. Anim. Sci. 2006. 84(E. Suppl.):E155–E165

INTRODUCTION

Broken-line regression analysis of nutrient dose response data 1) is relatively simple using standard nonlinear regression software, 2) provides a function that describes the response to nutrient dose across all dose levels, and 3) provides a break point estimate and standard error, interpreted as the nutrient requirement above which there is no significant change in the dependent variable. Linear, broken-line regression presumes that the response to nutrient dose is linear when it generally is not because the rate of change with nutrient dose decreases as the nutrient dose approaches its requirement. For some data sets, linear broken-line regression is sufficient, accounting for a substantial proportion of total variation and producing a visually satisfactory fit. For other data sets, the response is clearly curvilinear as the requirement is approached, and broken-line analysis using a straight-line, single-breakpoint model will underestimate the

requirement. For data such as these, an alternate model that includes a quadratic component is required.

Nutrient requirement experiments using swine are typically randomized complete block designs with gender, initial weight blocks, or both. Block effects are always included in the analysis of variance to decrease the error variance and to assess the significance of treatment effects more accurately; however, because of limitations of nonlinear model software, block effects generally are not used in requirement estimation analyses. The new SAS (SAS Inst., Inc., Cary, NC) procedure NLMixed fits nonlinear mixed models that allow for inclusion of block effects in broken-line regression analysis. This would seem most appropriate, given that gender and initial weight could affect the response to nutrient dose.

Our objective was to compare various models and SAS procedures NLIN and NLMixed in estimation of nutrient requirements. The general discussion of linear and nonlinear models in Littell et al. (1997) may suggest other model choices.

MATERIALS AND METHODS

The simplest broken-line is a 2 straight-line, single breakpoint with an increasing or decreasing slope at

¹Invited review. Presented at the American Society of Animal Science Midwest Section Meetings, Des Moines, IA, March 21–23, 2005.

²Corresponding author: krobbs@utk.edu

Received March 24, 2005.

Accepted August 11, 2005.

```

Data one; *Isoleucine requirement study (Parr et al. (2003));
input x y;
datalines;
0.38 459
0.42 602
0.46 687
0.50 695
0.54 729
0.58 725
;
data fill; *Generates multiple x values to facilitate graph of
predicted values;
  do x=.38 to .59 by .001; y=.; output; end;
run;
data one; set one fill; run;
proc sort data=one; by x;
proc nlin data=one; *straight broken-line;
parameters L=725 U=-2853 R=.5;
z1=(x<R)*(R-x);
model y = L + U*(z1);
output out=ppp p=predy;
run;

proc gplot;
  title2 '2 linear broken lines';
  goptions hpos=35 vpos=35 ftext=swiss;
  symbol1 v=dot c=black;
  symbol2 i=join v=none c=black;
  plot y*x predy*x/overlay;
run;

```

Source	DF	Sum of Squares	Mean Square	F Value	Approx Pr > F
Model	2	52792.2	26396.1	63.28	0.0035
Error	3	1251.3	417.1		
Corrected Total	5	54043.5			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits
L	716.3	11.7914	678.8 753.9
U	-2850.0	361.0	-3999.0 -1701.0
R	0.4669	0.00834	0.4404 0.4934

Figure 1. One-slope, straight broken-line analysis of swine growth response to varied levels of dietary isoleucine.

values of the dependent variable less than the breakpoint (requirement) and either a plateau at levels of the dependent variable above the breakpoint or a second line with increasing or decreasing slope. The single-slope model can be written as $y = L + U \times (R - x)$, where $(R - x)$ is defined as zero when $x > R$. The two-slope broken-line model would be $y = L + U \times (R - x) + V \times (x - R)$, where $(R - x)$ is defined as zero at values of $x > R$, and $(x - R)$ is defined as zero when $x < R$. We define parameters for the breakpoint x value (R), an asymptote for the first segment (L), and slopes for the 2 line segments (U , V). The code used for fitting these models in SAS Proc NLIN can be summarized as

```

Proc NLIN data = data set name;
Parameters L = initial value U = initial value V =
initial value R = initial value;

```

```

z1 = (x < R) * (R - x), where (R - x) is defined as
zero if (x > R);
z2 = (x > R) * (x - R), where (x - R) is defined as
zero if (x < R);
MODEL y = L + U * (z1) + V * (z2);
RUN.

```

To fit a one-slope model, simply delete any mention of V or $z2$. Or equivalently, to avoid use of the z indicator variables and to define explicitly the model in each segment, use the following code:

```

Proc NLIN data = data set name;
Parameters L = initial value U = initial value V =
initial value R = initial value;
if (x < R) then MODEL y = L + U * (R - x);

```

```
proc nlin data=one; *Quadratic broken line;
parameters L=716 V=-18000 R=.501;
z1=(x<R)*(R-x);
model y = L + V*(z1)*(z1);
output out=ppp p=predy ;
run;
```

Source	DF	Sum of Squares	Mean Square	F Value	Approx Pr > F
Model	2	53352.8	26676.4	115.87	0.0014
Error	3	690.7	230.2		
Corrected Total	5	54043.5			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
L	716.5	8.7683	688.6	744.4
V	-17557.1	4260.2	-31115.3	-3999.0
R	0.5010	0.0140	0.4564	0.5456

Approximate Correlation Matrix				
	L	V	R	
L	1.0000000	0.3445211	0.4990274	
V	0.3445211	1.0000000	0.9626595	
R	0.4990274	0.9626595	1.0000000	

Figure 2. One-slope, quadratic broken-line analysis of swine growth response to varied levels of dietary isoleucine.

else MODEL $y = L + U * 0$; use this if fitting the one-slope model;
 else MODEL $y = L + V * (x - R)$; use this if fitting the 2-slope model;
 RUN.

The SAS Proc NLMixed procedure is new, and it allows inclusion of random (block) components in the model. Because most swine nutrient requirement experiments are blocked by gender, initial BW, or both, it is prudent to include the variance associated with block in broken-line analysis. An example of SAS code to accomplish this when fitting a single-slope broken-line analysis is

```
Proc NLMixed data = data set name;
Parameters L= initial estimate U = initial estimate
R = initial estimate;
z1 = (x < R) * (R - x);
Model y ~ Normal (L + U * (z1) + block, errvar);
Random block ~ Normal (0, repvar) subject = rep;
Run.
```

In both NLIN and NLMixed model statements, the slope of the lines on either side of R can be linear or curvilinear. To estimate parameters for a quadratic model, one could simply use

```
Model y = L + U * (z1) * (z1), for NLIN;
Model y ~ Normal (L + U * (z1) * (z1) + block, errvar),
for NLMixed.
```

To assist in choosing an appropriate model, R^2 and adjusted R^2 values were calculated using the following

formulas because NLIN and NLMixed do not provide these.

$$R^2 = (CTSS - SSE)/CTSS$$

$$\text{Adjusted } R^2 = 1 - [SSE/(N - P - 1)]/[CTSS/(N - 1)].$$

The R^2 is the fraction of variation in the dependent variable explained by the model, and the adjusted R^2 adjusts for the number of parameters fitted in the model. Residual values were used to compute the necessary sums of squares for error (SSE), and corrected




Figure 3. The quadratic broken-line: $y = L + U * (R - x) * (R - x)$, where $(R - x)$ is zero at values of $x > R$, fitted to the isoleucine growth response data of Parr et al. (2003), ignoring blocks.

```

proc nlin data=one; *Quadratic function joined at plateau;
  parms a=.45 b=.05 c=-.0025;
  x0=-.5*b / c;          * Estimate join point;
  if x<x0 then
    model y=a+b*x+c*x*x;
  else
    model y=a+b*x0+c*x0*x0 ;
  if _obs_=1 and _iter_=. then do;
    plateau=a+b*x0+c*x0*x0;
    put / x0= plateau= ;
  end;
  output out=ppp predicted=predy;
run;

```

Source	DF	Sum of Squares	Mean Square	F value	Approx Pr > F
Model	2	53352.8	26676.4	115.87	0.0014
Error	3	690.7	230.2		
Corrected Total	5	54043.5			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
a	-3690.7	837.3	-6355.4	-1026.0
b	17592.7	3797.4	5507.6	29677.9
c	-17556.8	4260.2	-31114.8	-3998.8

Approximate Correlation Matrix			
	a	b	c
a	1.0000000	-0.9992419	0.9972852
b	-0.9992419	1.0000000	-0.9993841
c	0.9972852	-0.9993841	1.0000000

x0=0.5010236628 plateau=716.49537374

Figure 4. Alternate approach to one-slope, quadratic broken-line analysis of swine growth response to varied levels of dietary isoleucine.

total sums of squares (CTSS) were used as the baseline. The number of observations in this formula is represented by N, and for random-effect models, number of parameters (P) was increased by the levels associated with the random term. For example, adding block to the above models adds one parameter for each replicate.

RESULTS AND DISCUSSION

Parr et al. (2003) evaluated the isoleucine requirement of growing (25 to 45 kg) pigs and concluded that use of the straight broken-line model (Figure 1) seemed to underestimate the isoleucine requirement. Those researchers noted that the growth response to lower levels of isoleucine was curvilinear, and they suggested that the abscissa of the intersection of the straight broken-line with a quadratic regression curve was a more accurate estimate. An alternative approach is to use a quadratic function in broken-line analysis. This could be accomplished by fitting a broken-line model of the form $y = L + U * (R - x) + V * (R - x) * (R - x)$. This model would fit a quadratic curve to values of $x < R$, with a plateau (L) at values of $x > R$; however, fitting a quadratic requires 3, and preferably 4, data points below R. For the isoleucine data

set, this model would restrict R to isoleucine values $> 0.50\%$. To overcome this, we simply eliminated the linear component and used the model $y = L + U * (R - x) * (R - x)$, as shown in Figure 2. This model estimated the isoleucine requirement to be 0.50% (Figures 2 and 3). An alternative approach is to fit a quadratic curve and force it to join a plateau defined by solving for x when the first derivative of the quadratic is set to zero. The code to accomplish this is presented in Figure 4, a direct application of an example in the SAS NLIN documentation. This approach also provided an isoleucine requirement estimate of 0.50% . The 2 approaches are not mathematically identical, but they yielded the same estimate. The advantage of the latter approach is that the linear component of the quadratic can be included in the model. The disadvantage is that the variance associated with R cannot be estimated in NLIN (but this can be accomplished with the Estimate statement in NLMixed).

Portz et al. (2000) studied the protein requirement of growing fish. These data showed a linear decrease in growth at levels of protein greater than the requirement, thereby suggesting a model of the form $y = L + U * (R - x) * (R - x) + V * (x - R)$. This model fits a quadratic function at values of $x < R$ and a straight line at values of $x > R$. The code and output are presented in

```

Data one; * Protein requirement of fish (Portz et al. 2000);
input x y;
datalines;
18 159.26
21 203.72
24 200.15
27 192.39
30 189.02
;
proc sort data=one; by x;
proc nlin data=one; * 2-slope broken line - quadratic at x<R and linear
at x>R;
parameters L=200 U=-225 V=-2 R=21;
z1=(x<R)*(R-x);
z2=(x>R)*(x-R);
model y = L + U*(z1)*(z1) + V*(z2);
output out=ppp p=predy ;
run;
proc gplot;
  goptions hpos=35 vpos=35 ftext=swiss;
  symbol1 v=dot c=black;
  symbol2 i=join v=none c=black;
  plot y*x predy*x/overlay;
run;

```

Source	DF	Sum of Squares	Mean Square	F Value	Approx Pr > F
Model	3	1233.7	411.2	128.03	0.0649
Error	1	3.2120	3.2120		
Corrected Total	4	1236.9			

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
L	204.3	1.7016	182.7	225.9
U	-3.9464	1.5716	-23.9151	16.0222
V	-1.8550	0.4224	-7.2224	3.5124
R	21.3776	0.6974	12.5169	30.2384

Figure 5. Two-slope, quadratic broken-line analysis of fish growth response to varied levels of dietary protein.

Figures 5 and 6 and illustrate the flexibility of broken-line analysis.

Parr et al. (2003) also measured plasma urea N concentrations in pigs fed graded levels of isoleucine. These data proved difficult to fit. We used a simple one-slope straight, broken-line model and, for illustration, also used the “if, else” code to define the model (Figure 7). Note that the code includes one run that failed with the message that the Hessian was singular followed by parameter estimates that lack variance estimates. For data sets such as these, we find that one can most often solve the problem of arriving at a solution by including a bounds statement for the R parameter. This effectively restricts the iterative process to R values within the bounds statement. Inclusion of the bounds statement in the second run (Figure 7), even with the same initial parameter estimates, resulted in a successful solution and an estimated isoleucine requirement of 0.54%. Bounds statements can be included for any or all parameters, but our experience suggests that restricting parameter R can very often solve the problems associated with hard-to-fit data sets.

The “if, else” code for inputting the model is intuitively clear and may be preferable for some. It also




Figure 6. The quadratic broken-line: $y = L + U * (R - x) * (R - x) + V * (x - R)$, where $(R - x)$ is zero at values of $x > R$ and $(x - R)$ is zero at levels of $x > R$, fitted to the protein growth response data of Portz et al. (2000).


```

Data one; *Plasma urea N response to isoleucine (Parr et al. 2003);
input x y;
datalines;
0.38 12.34
0.42 11.48
0.46 10.94
0.50 8.86
0.54 9.80
;
data fill; do x=0.381 to 0.539 by 0.001; y=.; output; end; run;
data one; set one fill; run;
proc sort data=one; by x;
proc nlin data=one;
  parameters L=9 U=20 R=0.48 ;
  if (x le R) then model y = L + U*(R-x);
  else model y = L + U*0;
run;
proc nlin data=one;
  parameters L=9 U=20 R=0.48 ;
  bounds 0.47<R<0.53; * bounds statement included;
  if (x le R) then model y = L + U*(R-x);
  else model y = L + U*0;
run;

Without bounds statement:

NOTE: Convergence criterion met but a note in the log indicates
      a possible problem with the model.

NOTE: The (approximate) Hessian is singular.
      Parameter      Estimate      Std Error
      L              9.1376        0.5710
      U             19.2500        5.8111
      R              0.5403         .

With bounds statement included:

      Parameter      Estimate      Std Error
      L             9.5622         0.6225
      U            22.8332         6.9602
      R             0.5000         0.0355

```

Figure 7. Straight broken-line analysis of swine plasma urea nitrogen response to varied levels of dietary isoleucine.

allows greater flexibility in constructing models. As an example, we used the growth data of Parr et al. (2003) to fit a 3 straight-line, 2-breakpoint model (Figures 8 and 9). The model contains 6 parameters, thereby requiring a minimum of 7 data points. Thus, for illustration purpose, we added a seventh data point (0.59, 725). The model successfully fit the data and yielded breakpoints at 0.44 and 0.54.

Pigs used in the isoleucine requirement experiment of Parr et al. (2003) were blocked by gender and initial BW. Because it is reasonable to assume that the response to isoleucine dose will vary with initial BW and gender, it is statistically appropriate to account for this source of variation in broken-line analysis. Terms such as gender can be added to models as fixed or random effects. Here, we do not want to compare gender responses to isoleucine (the fixed-effect viewpoint), rather we want to estimate the overall response but account for group differences in the statistical estimation process. The SAS procedure NLMixed allows in-

clusion of random (block) components. The authors provided their complete data set, and the program we used for analysis using Proc NLMixed is presented in Figures 10 and 11. We evaluated 3 different mixed models: the straight broken-line with a random component included for parameter L; a quadratic broken-line with a random component included for parameter L; and the quadratic with random components included for both L and U. We also used Proc NLMixed to fit the quadratic model without any random components included. The solution is identical to that achieved with PROC NLIN, but this allowed calculation of the appropriate statistics to compare all models directly.

Output from NLMixed does not contain an estimate of R^2 , so additional code was included to make that calculation using observed and residual values; moreover, the code also calculates an adjusted R^2 . Unadjusted R^2 is simply the fraction of total variance accounted for by a model. Addition of parameters to a

```

Data one; *Isoleucine requirement study (Parr et al. 2003);
input x y; *added value of x=.59 allowed fit of 6 parameter model;
datalines;
0.38 459
0.42 602
0.46 687
0.50 695
0.54 729
0.58 725
0.59 725
;
data fill; *Generates multiple x values to facilitate graph of
predicted values;
do x=0.38 to 0.59 by 0.001; y=.; output; end;
run;
data one; set one fill; run;
proc sort data=one; by x;
proc nlin data=one; * 3 straight lines with 2 breakpoints;
parameters L=4 U=200 V=10 W=.01 R1=0.45 R2=0.55;
bounds 0.48>R1>0.42, 0.56>R2>0.5;
if (x le R1) then model y = L + U*x ;
else do; if (R2 ge x > R1) then model y = L+R1*U + V*(x-R1);
else model y = L + R1*U + (R2-R1)*V + W*(x-R2);
end;
output out=ppp p=predy ; *generates predicted values;
run;
proc gplot; * plots predicted curve overlaid on observations;
title2 '3 straight-line segments model';
goptions hpos=35 vpos=35 ftext=swiss;
symbol1 v=dot c=black;
symbol2 i=join v=none c=black;
plot y*x predy*x/overlay;
run;

```

Parameter	Estimate	Approx Std Error	Approximate 95% Confidence Limits	
L	-899.5	150.3	-2809.2	1010.2
U	3575.0	375.3	-1193.3	8343.3
V	525.0	187.6	-1859.1	2909.1
W	-2.51E-6	1501.1	-19073.1	19073.1
R1	0.4396	0.00690	0.3519	0.5272
R2	0.5406	0.1290	-1.0985	2.1798

Figure 8. Three straight-line, 2 breakpoint broken-line analysis of swine growth response to varied levels of dietary isoleucine.




Figure 9. A straight-line, 2-breakpoint broken-line model fitted to the isoleucine growth response data of Parr et al. (2003).

base model will necessarily increase R^2 ; however, the increase may not be statistically meaningful. The adjusted R^2 removes or corrects for the extra variation accounted for by parameter addition alone and allows direct comparison of models with differing numbers of parameters. Note that the code used to estimate the adjusted R^2 is slightly different for the model in which no random components are included (Figure 11).

The inclusion of a random component for parameter L allowed the model to account for the fact that pigs in the heaviest block would be expected to weigh more than pigs in the lightest block at all isoleucine levels. Similarly, inclusion of a random component for the slope (U) accounts for the fact that the growth response to increasing levels of isoleucine by animals in a block containing all gilts may be expected to be steeper than for animals in a block containing all barrows.

The NLMixed output includes “fit statistics” and a series of statistics for each parameter, including one titled “gradient.” These values can be used to assess and compare fitted models. In evaluating output, one

A

```

data one;
  input Rep      Sex$ @;
  do x=0.38,      0.42,0.46,0.50,0.54,0.58;
    input y @; output;
  end;
datalines;
1      Ba      533      686      819      757      691      667
2      Ba      452      543      657      743      757      721
3      Gi      433      648      561      560      657      651
4      Gi      418      530      712      721      809      862
;
proc print;run;
data fill;
  do rep=1 to 4;
    do x=0.38 to 0.58 by 0.001; y=.; output;end;end; run;
data one; set one fill; run;
proc sort data=one; by rep; run;
*** Straight broken line with a random component included for parameter
1;
proc nlmixed data=one maxiter=1000;
parameters L=725 U= 130 R=0.5;
z1=(x<R)*(R-x);
model y ~Normal( L + U*(z1) + lvar, errvar);
random lvar~Normal(0, repvar) subject=rep out=blups;
predict L + U*(z1) + lvar out=ppp ;run;
ods output dimensions=ddd;
data ppp; set ppp;
  resid=y-pred; run;

**** run this code to get an R-square and adjusted R-square;
proc means noprint data=ppp; where y ne .;
  var y resid;
  output out=mmm uss=totss sserr css=ctotss rssl n=nobs; run;
data ddd; set ddd; if descr='Parameters'; run;
proc sort data=blups; by effect;
proc means noprint data=blups; by effect;
  var estimate;
  output out=nnn n=nlevels; run;
proc means noprint;
  var nlevels;
  output out=nnn2 n=nrand sum=dfrand; run;
data mmm; merge mmm ddd nnn2;
  rsquare = (ctotss-sserr)/ctotss;
  adjrsquare = 1 - (sserr/(nobs-value-dfrand+nrand-1)) / (ctotss/(nobs-
1)); run; proc print;run;

```

B

```

**** Run this code to get plot of fitted model;

proc sort data=ppp; by x;
data predplot; set ppp;
  if rep=1 then do; pred1=pred; y1=y; end;
  if rep=2 then do; pred2=pred; y2=y; end;
  if rep=3 then do; pred3=pred; y3=y; end;
  if rep=4 then do; pred4=pred; y4=y; end;
proc gplot data=predplot;
  title2 'Straight broken line with LVAR';
  goptions hpos=35 vpos=35 ftext=swiss;
  symbol1 v=dot c=black i=none;
  symbol2 v=square c=blue i=none;
  symbol3 v=triangle c=red i=none;
  symbol4 v=x c=green i=none;
  symbol5 i=join v=none c=black;
  symbol6 i=join v=none c=blue;
  symbol7 i=join v=none c=red;
  symbol8 i=join v=none c=green;
  plot (y1 y2 y3 y4)*x . (pred1 pred2 pred3 pred4)*x/overlay;
run;

```

Figure 10. Nonlinear, mixed model broken-line analysis of swine growth response to varied levels of dietary isoleucine.


```

*** Quadratic broken line with a random component included for
parameter 1;

proc nlmixed data=one maxiter=1000;
parameters L=725 U=0.13 R=0.5;
bounds 0.4<R<0.6;
z1=(x<R)*(R-x);
model y ~Normal( L + U*(z1*z1) + lvar, errvar);
random lvar~Normal(0, repvar) subject=rep out=blups;
predict L + U*(z1*z1) + lvar out=ppp ;
ods output dimensions=ddd; run;

*** Quadratic broken line, allowing parameter U (slope) to have random
component;

proc nlmixed data=one maxiter=1000;
parameters L=725 U=-2000 R=0.5;
bounds 0.4<R<0.54;
z1=(x<R)*(R-x);
model y ~Normal( L + (U+Ur)*(z1*z1) + lvar, errvar);
random Ur lvar~Normal([0,0],[uvar,0, repvar]) subject=rep out=blups;
predict L + (U+Ur)*(z1*z1) + lvar out=ppp ;
ods output dimensions=ddd;
run;

**** Quadratic model without random (block) effects ****;

proc nlmixed data=one maxiter=1000;
parameters L=716 U= -17500 R=0.5;
bounds 0.4<r<0.6;
z1=(x<R)*(R-x);
model y ~Normal( L + U*(z1)*(z1), errvar);
predict L + U*(z1)*(z1) out=ppp ;
run;
***** run this code to get an R-square with no random (block) effects
in model;
data ppp; set ppp;
resid=y-pred; run;
proc means noprint data=ppp; where y ne .;
var y resid;
output out=mmm uss=totss sserr css=ctotss rssl n=nobs; run;
data ddd; set ddd; if descr='Parameters'; run;
data mmm; merge mmm ddd ;
rsquare = (ctotss-sserr)/ctotss;
adjrsquare = 1 - (sserr/(nobs-value-1)) / (ctotss/(nobs-1)); run;
proc print; run;

```

Figure 11. Alternate nonlinear mixed models.

should first examine the gradient values for each parameter estimate. The iterative process used to estimate parameters involves minimizing a complex function, and the minimum is where the gradients (i.e., partial derivatives of the function) are equal to zero. A large gradient suggests the parameter estimate is not located at the minimum and, thus, could be changed to produce a better model fit. If any gradient is much above 0.001, one should adjust the initial parameter estimates and rerun the program. Fadel (2004) discussed a strategy to identify good initial pa-

rameter estimates. For the output from the straight-line mixed model (Figure 12), note that all gradient values were small, indicating that the parameter estimates were very near the minimum point on the estimation curve, but occasionally exceeded 0.001. Attempts should be made to improve these values, but with limited data, it may not always be possible.

For the Akaike Information Criterion and -2 log likelihood fit statistics, smaller values indicate better fit. We use the approximation that for each parameter added, the -2 log likelihood should drop by 5 units

Straight broken line with a random component for parameter L included in the model:								
				Fit Statistics				
				-2 Log Likelihood		288.3		
				AIC (smaller is better)		298.3		
				AICC (smaller is better)		301.6		
				BIC (smaller is better)		295.2		
Parameter Estimates								
Parameter	Estimate	Standard Error	DF	t Value	Pr > t	Alpha	Gradient	
L	708.87	42.0436	3	16.86	0.0005	0.05	-0.00011	
U	-5106.18	850.01	3	-6.01	0.0092	0.05	-0.00235	
R	0.4350	0.006997	3	62.16	<.0001	0.05	0.000297	
errvar	2090.39	357.78	3	5.84	0.0100	0.05	-0.0066	
repvar	6548.07	0.3705	3	17673.3	<.0001	0.05	0.00022	
totss	sserr	ctotss	rssl	rsquare	adjrsquare			
10467704	99935.22	343298	99935.22	0.70890	0.55364			

Figure 12. Example output of nonlinear mixed model broken-line analysis of swine growth response to varied levels of dietary isoleucine.

(based on the χ^2 distribution) before concluding that parameter addition significantly improved the model. Examination of the output obtained for the various models (Figure 13) shows that the quadratic with a random component for parameter L had the smallest log likelihood fit statistic. The -2 log likelihood was significantly better than that achieved for the straight-line model, but only slightly better than for the other models. The adjusted R^2 was slightly greater than that achieved with the quadratic model without any random components included, thereby suggesting that this model best described the response. As expected, the unadjusted R^2 increased across models as parameters were added, illustrating the weakness of this measure of model fit.

The estimated isoleucine requirement for the quadratic model with a random component for parameter L was 0.52%. This fitted model and observed values are graphically presented in Figure 14. In this graph, the height differences among replicates (blocks) repre-

sents the variance associated with block for parameter L.

For these data, we observed only a small gain in information by including random components in the broken-line model; however, we would expect greater gain in information for larger data sets or data sets with higher block-associated variances. Thus, we recommend that such models be tested in broken-line analysis of nutrient response data obtained from experimental designs that include blocks.

In our limited experience with Proc NLMixed, we have observed that it is especially sensitive to initial parameter estimates, so one may need to attempt several runs with different initial parameter estimates before achieving a solution. On occasion, we also have noted that solutions may be at local rather than the global minimum, so critical evaluation of the parameter estimates and careful inspection of graphical output is also recommended. It may prove helpful to obtain initial parameter estimates using NLIN, which

Model ¹	Estimated requirement, %	AIC ²	-2 log likelihood fit statistic	R ²	Adjusted R ²
Quadratic broken line with no block effects included	0.501 ± 0.009	282.4	274.4	0.622	0.489
Straight broken line with a random component for parameter L	0.435 ± 0.007	298.2	288.3	0.709	0.554
Quadratic broken line with a random component for parameter L	0.517 ± 0.010	283.4	273.4	0.695	0.532
Quadratic broken line with random components for parameters L and U	0.472 ± 0.006	288.3	276.3	0.724	0.422

¹The quadratic broken-line model is $y = L + U * (R - x) * (R - x)$; the straight broken-line model is $y = L + U * (R - x)$, where L is the ordinate, R is the abscissa of the breakpoint, and the value $R - x$ is zero at values of $x > R$.

²Akaike information criterion.

Figure 13. Results of applying various mixed models in broken-line analysis of swine growth response to varied levels of dietary isoleucine.




Figure 14. The quadratic broken-line: $y = L + U * (R - x) * (R - x) + LVAR$, where $(R - x)$ is zero at values of $x > R$, fitted to the isoleucine growth response data of Parr et al. (2003). Parameter LVAR represents the random (block) effect and produces different gain lines (shown by solid and various dashed lines) for the respective blocks.

displays fewer tendencies to converge at local minima, and then use these parameter estimates in Proc NLMixed.

IMPLICATIONS

The SAS procedures NLIN and NLMixed allow for considerable flexibility in model development and provide an objective estimate of the nutrient requirement and its associated variance. Straight-line or quadratic models can be easily fit. The NLIN procedure is the preferred procedure for analyzing nutrient response data from experiments that are not blocked, but when analyzing data from experiments with blocks, one should also evaluate models containing random (block) effects for both parameter L and U to ensure that one or more blocks does not bias the outcome. Complete SAS code for accomplishing this is presented.

LITERATURE CITED

- Fadel, J. G. 2004. Technical Note: Estimating parameters of nonlinear segmented models. *J. Dairy Sci.* 87:169–173.
- Littell, R. C., P. R. Henry, A. J. Lewis, and C. B. Ammerman. 1997. Estimation of relative bioavailability of nutrients using SAS procedures. *J. Anim. Sci.* 75:2672–2683.
- Parr, T. M., B. J. Kerr, and D. H. Baker. 2003. Isoleucine requirement of growing (25 to 45 kg) pigs. *J. Anim. Sci.* 81:745–752.
- Portz, L., C. T. D. S. Dias, and J. E. P. Cyrino. 2000. Regressao segmentada como modelo na determinacao de exigencias nutricionais peixes. *Scientia Agricola* 57:601–607.